Algorithm AS 32

The Incomplete Gamma Integral

By G. P. BHATTACHARJEE

Department of Mathematics, I.I.T., Kharagpur (India)

LANGUAGE

ASA Standard Fortran

DESCRIPTION AND PURPOSE

For given values of $x(\geqslant 0)$, $p(>0)$ and the complete gamma function $\Gamma(p)$, the function subprogram computes the incomplete gamma ratio defined by

$$I(x,p) = \frac{1}{\Gamma(p)} \int_0^x \exp(-t)\, t^{p-1}\, dt.$$

Numerical method

The subprogram uses the series expansion (c.f. Pearson, 1922) viz.

$$I(x,p) = \frac{\exp(-x)\, x^p}{\Gamma(p+1)} \left[1 + \sum_{r=1}^{\infty} \frac{x^r}{(p+1)(p+2)\dots(p+r)}\right]$$

for $p \leqslant x \leqslant 1$ and also for $x < p$. The series is terminated when the contribution to the series is not greater than the value of ACU defined in the first arithmetic statement of the subprogram. For all other cases, the continued fraction expansion (c.f. Abramowitz and Stegun, 1965) viz.

$$I(x,p) = 1 - \frac{\exp(-x)\, x^p}{\Gamma(p)} \left[\frac{1}{x+}\, \frac{1-p}{1+}\, \frac{1}{x+}\, \frac{2-p}{1+}\, \frac{2}{x+} \dots\right]$$

is used. The ratio is calculated by successive iteration and the process is terminated when the relative difference between two successive iterations is not greater than ACU.

STRUCTURE

FUNCTION GAMAIN (X,P,G,IFAULT)

Formal parameters

X	Real	input:	the value of the upper limit x.
P	Real	input:	the value of the parameter p
G	Real	input:	the natural logarithm of $\Gamma(p)$.
IFAULT	Integer	output:	a fault indicator, equal to: 1 if $p \leqslant 0$, 2 if $x < 0$ and 0, otherwise.

Auxiliary algorithm

For the natural logarithm of $\Gamma(p)$ any standard algorithm, such as ACM Algorithm 291, may be used.

Time

The time depends on the relative values of x and p, but on an average 550 IBM 1620 instructions are executed, using a version without logical *IF* statements.

Accuracy

The accuracy of the result normally should not be less than the value of ACU defined as the first arithmetic statement of the subprogram. But due to truncation and rounding off errors the last digit may not be correct.

Thanks are due to the referee for suggesting improvements to the original version of the program.

References

Abramowitz, M. and Stegun, I. A. (1965). *Handbook of Mathematical Functions.* New York: Dover.

Pearson, K. (1922). *Tables of the Incomplete Gamma Function.* London: Cambridge University Press.

Pike, M. C. and Hill, I. D. (1966). Algorithm 291: Logarithm of the gamma function. *Comm. A.C.M.*, 9, 684.

```
      FUNCTION GAMAIN(X,P,G,IFAULT)
C
C        ALGORITHM AS 32 J.R.STATIST.SOC. C. (1970) VOL.19 NO.3
C
C        COMPUTES INCOMPLETE GAMMA RATIO FOR POSITIVE VALUES OF
C        ARGUMENTS X AND P.  G MUST BE SUPPLIED AND SHOULD BE EQUAL TO
C        LN(GAMMA(P)).
C        IFAULT = 1 IF P.LE.0 ELSE 2 IF X.LT.0 ELSE 0.
C        USES SERIES EXPANSION IF P.GT.X OR X.LE.1, OTHERWISE A
C        CONTINUED FRACTION APPROXIMATION.
C
      DIMENSION PN(6)
C
C        DEFINE ACCURACY AND INITIALIZE
C
      ACU=1.0E-8
      OFLO=1.0E30
      GIN=0.0
      IFAULT=0
C
C        TEST FOR ADMISSIBILITY OF ARGUMENTS
C
      IF(P.LE.0.0) IFAULT=1
      IF(X.LT.0.0) IFAULT=2
      IF(IFAULT.GT.0.OR.X.EQ.0.0) GO TO 50
      FACTOR=EXP(P*ALOG(X)-X-G)
      IF(X.GT.1.0.AND.X.GE.P) GO TO 30
C
C        CALCULATION BY SERIES EXPANSION
C
      GIN=1.0
      TERM=1.0
      RN=P
   20 RN=RN+1.0
      TERM=TERM*X/RN
      GIN=GIN+TERM
      IF(TERM.GT.ACU) GO TO 20
      GIN=GIN*FACTOR/P
      GO TO 50
```

```
C
C        CALCULATION BY CONTINUED FRACTION
C
   30 A=1.0-P
      B=A+X+1.0
      TERM=0.0
      PN(1)=1.0
      PN(2)=X
      PN(3)=X+1.0
      PN(4)=X*B
      GIN=PN(3)/PN(4)
   32 A=A+1.0
      B=B+2.0
      TERM=TERM+1.0
      AN=A*TERM
      DO 33 I=1,2
   33 PN(I+4)=B*PN(I+2)-AN*PN(I)
      IF(PN(6).EQ.0.0) GO TO 35
      RN=PN(5)/PN(6)
      DIF=ABS(GIN-RN)
      IF(DIF.GT.ACU) GO TO 34
      IF(DIF.LE.ACU*RN) GO TO 42
   34 GIN=RN
   35 DO 36 I=1,4
   36 PN(I)=PN(I+2)
      IF(ABS(PN(5)).LT.OFLO) GO TO 32
      DO 41 I=1,4
   41 PN(I)=PN(I)/OFLO
      GO TO 32
   42 GIN=1 0-FACTOR.GIN
C
   50 GAMAIN=GIN
      RETURN
      END
```

Algorithm AS 33

Calculation of Hypergeometric Sample Sizes

By F. B. Leech

Rothamsted Experimental Station, Harpenden, Herts

Language

ASA Standard Fortran

Purpose

Solution, in n, of $P = \{(N-m)!\,(N-n)!\}/\{N!\,(N-m-n)!\}$ for given values of P, N and $p = m/N$.

The program was originally written in connection with a paper on innocuity testing of inactivated foot-and-mouth disease vaccines (Anderson *et al.*, 1970). In this context, a sample of n cm^3 from a vaccine batch of N cm^3 is tested for the presence of residual active virus. Batches giving any reaction are rejected and the theoretical problem is to select a size of sample such that when the test is negative, the amount, if any, of active virus in the batch must be small. In this context, the theory involves